Fully Developed Flow in a Circular Tube

Equation (5-117) is the appropriate *x*-momentum equation for fully developed laminar flow in a circular tube. This differential equation is entered into Maple:

```
> restart;
> ODE:=-dpdx+mu*diff(r*diff(u(r),r),r)/r;
```

$$ODE := -dpdx + \frac{\mu\left(\left(\frac{d}{dr}\mathrm{u}(r)\right) + r\left(\frac{d^2}{dr^2}\mathrm{u}(r)\right)\right)}{r}$$

Maple will identify the solution using dsolve command:

```
> us:=dsolve(ODE);
```

$$us := \mathrm{u}(r) = \frac{dpdx\ r^2}{4\,\mu} + \_C1\ \ln(r) + \_C2$$

or:

$$u = \frac{r^2}{4\,\mu}\frac{dp}{dx} + C_1 \ln\left(r\right) + C_2 \tag{5-146}$$

The constants of integration, C_1 and C_2, are obtained by enforcing the boundary conditions. The velocity must remain bounded at the center of the tube; substituting $r = 0$ into Eq. (5-146) leads to:

$$u_{r=0} = C_1 \ln\left(0\right) + C_2 \tag{5-147}$$

which can only be bounded if $C_1 = 0$. This result is substituted into the Maple solution:

```
> us:=subs(_C1=0,us);
```

$$us := \mathrm{u}(r) = \frac{dpdx\ r^2}{4\,\mu} + \_C2$$

The no slip condition is enforced at the tube surface:

$$u_{r=D/2} = 0 \tag{5-148}$$

where D is the diameter of the tube. The value of C_2 that satisfies Eq. (5-148) is determined symbolically and substituted into the solution in Maple:

```
> us:=subs(_C2=solve(rhs(eval(us,r=D/2))=0,_C2),us);
```

$$us := \mathrm{u}(r) = \frac{dpdx\ r^2}{4\ \mu} - \frac{dpdx\ \mathrm{D}^2}{16\ \mu}$$

Therefore, the fully developed velocity distribution in a round tube is:

$$u = \frac{r^2}{4\,\mu}\frac{dp}{dx} - \frac{D^2}{16\,\mu}\frac{dp}{dx} \tag{5-149}$$

The pressure gradient will be negative in the direction of flow. Therefore, Eq. (5-149) is a parabolic velocity distribution that is zero at the wall and reaches a maximum value at the centerline. The mean velocity is defined in Section 5.1.2 as the single velocity that represents the flow rate:

$$u_m = \frac{\dot{V}}{A_c} \tag{5-150}$$

The mean velocity can be obtained by integrating the velocity distribution over the tube cross-sectional area:

$$u_m = \frac{1}{A_c}\int_{A_c} u\, dA_c \tag{5-151}$$

The integral in Eq. (5-151) is carried out in Maple:

```
> um:=int(rhs(us)*2*Pi*r,r=0..D/2)/(Pi*D^2/4);
```

$$um := -\frac{dpdx\ \mathrm{D}^2}{32\ \mu}$$

Therefore:

$$\boxed{u_m = -\frac{D^2}{32\,\mu}\frac{dp}{dx}} \tag{5-152}$$

Equation (5-152) is substituted into Eq. (5-149) in order to express the velocity distribution in terms of the mean velocity:

$$\boxed{u = 2\,u_m\left[1 - \left(\frac{2\,r}{D}\right)^2\right]} \tag{5-153}$$

The friction factor for an internal flow is defined in Section 5.1.2 as:

$$f = -\frac{dp}{dx}\frac{2\,D}{\rho\,u_m^2} \tag{5-154}$$

Substituting Eq. (5-152) into Eq. (5-154) leads to:

$$f = \underbrace{\frac{32\,\mu\,u_m}{D^2}}_{-\frac{dp}{dx}}\frac{2\,D}{\rho\,u_m^2} \tag{5-155}$$

or

$$\boxed{f = \frac{64}{Re_D}} \tag{5-156}$$

which is consistent with the friction factor for fully developed, laminar flow in a circular tube presented in Section 5.2.3.